
Assembly. Machine Language. For many people these are still riddles without solutions. But many experienced programers swear that Assembler is THE language to use, not to mention being easier and more flexible than Basic. But why? Well, first of all Assembler-instructions are very small and don't actually do much by themselves. However, their combined effect can be quite impressive. To understand these instructions, we first have to learn to count in binary and hexadecimal, since most operations are easier to understand when expressed in this form.

BITS AND BYTES

First, let's talk about bits. A bit is the smallest information unit and it can have one of two states: set and clear, or set and reset if you prefer. In other words, your choices are 1 and 0. Eight bits make one byte, which means that with a little simple math, we find that there are 2 to the 8th power, or 256, possible values.

HEXADECIMAL AND DECIMAL

When you count, you do so starting at zero or one, working your way up to ten. Take a look at the following sequence of numbers:

0,1,2,3,4,5,6,7,8,9

You will notice that to represent any number, you need only one digit, until you hit 10, at which point you need two digits to describe your number. Keep counting and you eventually need to add even more digits as you reach 100, 1000 and so on.

In Hexadecimal (or "Hex"), it works a little differently. Consider the following sequence:

0,1,2,3,4,5,6,7,8,9,A,B,C,D,E,F

In this sequence we have 16 values, ranging from 0 to F. But what do the letters A through F mean? Well, since there are no single digit characters in the English language that represent 10 through 15 in a natural way, we are forced to choose letters of the alphabet, and so, A through F became the standard. To further differentiate a decimal number from a Hex number, most coders use a dollar sign '$' in front of a Hex number.

As with decimal, we eventually have to add another digit to our count as we continue to increase in value. With decimal, we did this at 10. In Hex, we add a digit when we hit 16.

In decimal, you could say that a number WXYZ is W*1000 + X*100 + Y*10 + Z. So, the value 1234 would be 1000 + 200 + 30 + 4 of course.

In Hex, you would thus say that a number ABCD is A*4096 + B*256 + C*16 + D. As an example, $1234 (Hex) would be 1*4096 + 2*256 + 4*16 + 4, or 4676, when expressed in decimal.

In each case, each of the four digits in the numbers above is just that, a single digit, either 0 through 9 for the decimal system, or 0
through F for Hexadecimal.

As mentioned above, one byte can hold any value from 0 to 255 ($00 to $FF). To hold larger numbers, like $1234 in the above example, we simply use two bytes, each storing two digits from the number. The byte representing the right most digits is called the "LSB" or "Least Significant Byte", while the byte representing the digits to the right is called the "MSB" or "Most Significant Byte."

ASSEMBLY LANGUAGE - STEP BY STEP

THE REGISTERS:

In Assembly, your primary activity will be moving data back and forth, manipulating bits and bytes, and making comparisons and jumps throughout your program.

Inside every Commodore 64 is a MOS 6510 processor, the big brother to the 6502 that is used in the VIC-20 and most disk drives. This processor has three registers that can be used for many purposes. They are, the accumulator, denoted here as ".A", the X Index Register, denoted as ".X", and the Y Index Register, denoted as ".Y".

Each register is one byte in size, hence each holds a value from $00 to $FF (Hex).

THE MAIN STORAGE:
In addition to the registers, the 6510 has access to 65536 bytes of User-Programmable memory. Of course, every Commodore 64 comes fully loaded with a full 64K, which is enough to suit almost any need. Each individual byte has an address within the range of $0002 to $FFFF, with the very first two bytes taken by the processor for it's on-board parallel port.

THE FIRST COMMANDS:

Now it is time for you to load and start a machine language monitor. If you have an Action-Replay, Final Cartridge, Action Gear, Nordic Power or anything comparable, you can use the command MON to jump from the basic interpreter into the Cartridge's internal Machine Language Monitor.

Now lets talk about the most important commands: LDA, STA and JMP.

LDA:

LDA is an abbreviation (Mnemonic) for Load Accumulator. We use LDA to load a one-byte value into .A. The simplest LDA command is LDA #Value. As an example, LDA #$01 flat-out loads the value 1 into .A,

LDA #$02 the value 2, and so on for any value $00 through $FF. Note that the "#" sign is required, to specify "immediate" mode.


STA:

STA is the abbreviation for Store Accumulator. With STA we store the contents of .A to someplace in main memory (or perhaps, into an I/O chip like the SID). The contents of .A are left unchanged after the store operation. The simplest STA command is STA Address. For example, STA $3000 would store the contents of .A into location $3000 in main memory. STA $0400 would store to $0400, which is the start of your 40 column display.

JMP: 

JMP is the abbreviation for Jump. JMP is the 6510's "GOTO" command. Every address of the main storage can contain data or programs. With JMP you order the processor to stop what it's doing, move to a new place in your program or perhaps into the Operating System, and begin executing. Normally, you write the JMP command as JMP $nnnn where $nnnn is a location in the C64's main memory.

THE FIRST SMALL PROGRAM:

For our first small program we only need the mentioned 3 commands and 2 important storage locations you should keep in mind: $D020 and $D021. $D020 is the control byte for the screen's border color, while $D021 controls the background color of the text-portion of the screen.

And now, on to the ML Monitor. Below is a display typical of what you will see when you start your ML Monitor (either with "mon", "m or shift-N" on the C128 in C128 Native Mode, or by using a menu within your utility cartridge)

MON
B*
ADDR AR XR YR SP 01 NV-BDIZC
.; FFFF 00 00 00 F8 37 00000010

The first line means "Address" (where the computer was executing at when the monitor was called), .A .X and .Y registers, Stack Pointer, the value of location $0001, and the values of the 6510's various status flags (more on these last three items later)

Let's try our first program. Try the following few lines of code. Enter each line without the leading period (your ML Monitor will usually put it there for you), and press return at the end of each line. Depending on the ML Monitor you are using, the line may either be accepted as-is, corrected in some way, or altered to include such information as the hexadecinal values that make up the code you've entered.

As you enter each line, the ML Monitor will print the address of the next instruction and position the cursor to the right of that address, sort of like an "auto-line-number" feature.

.A 2000 LDA #$00
.A 2002 STA $D020
.A 2005 STA $D021
.A 2008 LDA #$01
.A 200A STA $D020
.A 200D STA $D021
.A 2010 JMP $2000
.A 2013 (just press Return)


What this program does:
2000 Load .A with the value $00
2002 Write contents of .A to the VIC chip's Border Color Register (#$00 was loaded into .A on the previous line, so this turns the border black)
2005 Write contents of .A (still #$00) to the VIC chip's Background Color Register. This turns the background black as well.
2008 Load .A with the value $01
200A Write contents of .A to the Border Color Register. Since we just loaded #$01 into .A, the border will now turn white.
200D Write contents of .A to the Background Color Register (turns the background white)
2010 Jump to memory location $2000 and continue. Since we are JMP'ing back to the beginning of the program, we created an "infinite" loop.

You may run this program by entering the command G 2000 at the next available ML Monitor prompt (if it produces one, usually a ".")

AND IN BASIC?

This is how the program might look if written in BASIC:

10 poke 53280,0
20 poke 53281,0
30 poke 53280,1
40 poke 53281,1
50 goto 10

In both cases we simply make the border and background colors flicker wildly (black to white, over and over). You'll notice the BASIC version runs considerable slower, as the screen will fill with stripes instead of thin, broken lines.

The Basic-program does look smaller, doesn't it? Actually, it's larger. The custom crafted machine code takes a mere 19 bytes of space (from $2000 to $2012), while the BASIC version hogs a whopping 52 bytes! Part of the reason for this is that the numbers 53280 and 53281 are actually being spelled out byte for byte in the program, while the numbers $D020 and $D021 in our ML example are being stored as binary numbers, taking only two bytes each.

In addition, BASIC is full of things like "line links" and line number values. All of these generally make BASIC slow and bloated in comparison.

CONCLUSION

As you can see, Machine Language really isn't that complex.  Just as learning to program in BASIC seemed complicated at first, you simply have to break the ice and start with something small. Once you've gotten your feet wet, you'll see it's really pretty easy to learn.

For those who want to get into Machine Language now, without waiting for future articles and hints, at least start by picking up a pocket calculator that features Hexadecimal and Binary conversion keys. Some calculators in the Casio FX series feature these, and they are quite handy.
Assembler Course Part 2
by Wanja Gayk

Hi everyone! I hope you got along well with the first part and that you still want to go on. This time we're going to really start off. So get your monitors warmed up! Let's recall the program from last course in which we switched the color of frames and the background from black to white and back again. It looked like this (you had to start it with SYS 8192 from BASIC or G 2000 from your ML monitor):

.A2000 LDA #$00        ; value of black into the accumulator,
.A2002 STA $D020    ; loads value of accumulator into register for frame colors,
.A2005 STA $D021    ; loads value of accumulator into register for background colors,
.A2008 LDA #$01        ; value of white into the accumulator,
.A200A STA $D020    ; loads value of accumulator into register for frame colors,
.A200D STA $D021    ; loads value of accumulator into register for background colors,
.A2010 JMP $2000    ; jump back to $2000

Of course there's a much easier way to do this - because you have various registers: accumulator, X and Y. Load and Store commands exist also for the X and Y registers. They are called LDX, STX and LDY, STY. LDX means LoaD X, STX means STore X. LDY and STY, respectively, are short for LoaD Y and STore Y. We can use them in the same simple way we used LDA and STA earlier on. Here's a short program:

.A2000 LDX #$00
; loads $00 (black) into the X register,
.A2002 LDY #$01
; loads $01 (white) into the Y register,
.A2004 STX $D020
; value of X register into the register for frame colors,
.A2007 STX $D021
; value of X register into the register for background colors,
.A200A STY $D020
; value of Y register into the register for frame colors,
.A200D STY $D021
; value of Y register into the register for background colors,
.A2010 JMP $2004
; jump back to $2004

Now, where's the difference? Seemingly there is none - the program even has the same length - but it operates faster. That's because there are fewer commands in its loop to change the color from black to white. Compare: the first program has seven commands (including JMP) in its closed loop. Program No.2 has only five of them - by the way, it is one of the fastest programs possible to implement that effect. Let's now try a different approach to the program, so that we can learn some new commands. We'll stick to the same effect not to loose track. We will focus exclusively on the X register, and the commands we need now are INX and DEX.

INX/INY

INX stands for INcrement X. It increases the value of X by one. If the value in the X register is $fd (253), INX will raise it to $fe (254). If we use another INX the value will increase to $ff (255). But we know that the accumulator, X and Y register only have a capacity of one byte, which is only fit for representing a value of $00 up to $ff. When trying to increase a value of $ff once more the processor will start again at $00, so there's no problem. In the 6510, the same command is also available for the Y register (INY for INcrement Y), but not for the accumulator.

DEX/DEY

DEX stands for DEcrement X and has the opposite effect of INX. It decreases the value of the X register by one. If the value in the X register is $02, DEX will lower it to $01. If we use DEX once more $01 will be decreased to $00. Again, the value will jump back and the processor will start counting at the other end of the scale if we try to decrease it once more and a value of $00 will be turned into $ff. The corresponding command for the Y register is DEY (DEcrement Y).

Let's write a short program using these two commands:

.A2000 LDX #$00
; loads $00 into the X register (black),
.A2002 STX $D020
; X register into frame color,
.A2005 STX $D021
; X register into background color,
.A2008 INX

; increases value of X register by one ($00 turns into $01 - white),
.A2009 STX $D020
; X register into frame color,
.A200C STX $D021
; X register into background color,
.A200F DEX

; decreases value of X register by one ($01 turns into $00 - black),
.A2010 JMP $2002
; and we start again at $2002

This program is of the same length as the previous ones and has the same effect - but it only works as fast as the first one because there are seven commands in the loop that have to be executed, with INX and DEX taking up as much time as LDA - just to tell you. Now, wouldn't it be nice if we could increase or decrease values directly in the main storage? Of course this is also possible by using INC and DEC. INC and DEC work the same way as INX and DEX, with the difference that affect the contents of one single storage location, rather than the contents of the X or Y register. INC means INCrement and DEC DECrement. The next program shows the practical application:

.A2000 LDA #$00
.A2002 STA $D020
; frame color black,
.A2005 STA $D021
; background color black,
.A2008 INC $D020
; adds one to frame color (= $01 - white),
.A200B INC $D021
; adds one to background color (= $01 - white),
.A200E DEC $D020
; subtracts one from frame color (= $00 - black),
.A2011 DEC $D021
; subtracts one from background color (= $00 - black),
.A2014 JMP $2008
; jump back to $2008

This program is a little longer than the previous ones due to the fact that both storage cells first have to be initialized with $00 in order to be able later on to change between $00 and $01. Even though the program's loop consists of only five commands this program is slower than any of the others, since INC as well as DEC take up six clock cycles each. LDA, LDX and LDY in comparison use only two, STA, STX and STY take up four clock cycles, INX/INY and DEX/DEY also use only two while JMP swallows up three clock cycles. These numbers are only valid for the commands already introduced as there are different variants of LDA and STA commands. This leads us to our last programs for today, for which we'll need two new commands.

LDA (X-indexed) and LDA (Y-indexed)

LDA $3000, X is a great command to read tables. It loads the content of the storage cell $3000 plus X into the accumulator. The X register is used as, well, an index. If X = $00, then the content of the storage cell $3000 will be read, if X = $35 the value will come from the storage cell $3035, and so on. The same thing works also for the Y register - with the command LDA $3000, Y.

BEQ (Branch if EQual), BNE (Branch if Not Equal) and the ZERO-FLAG

BEQ is a command that effects a jump similar to JMP, but only if certain conditions are fulfilled, that is, if the so-called ZERO-FLAG is set. BNE also jumps, but only if the ZERO-FLAG is clear. The ZERO-FLAG is a processor bit with which the value of a command resulting in zero is marked. That is, if you execute a DEX while a value of $01 is in the X register, the result will be $00, and therefore the ZERO-FLAG will be set. This is also true if DEC results in $00 or if $ff is turned into $00 by INX, INY or INC. The ZERO-FLAG will also be set if LDA #$00, LDX #$00 or LDY #$00 are executed. If you load a value of $00 from main storage into the accumulator with an LDA, LDX, or LDY, the ZERO-FLAG will be set. It is cleared if the result of a command does not equal zero. The following short program is to clarify the way it works:

.A2000 LDX #$04
; loads X register with $04,
.A2002 STX $D020
; value from X register into frame color,
.A2005 DEX

; decreases X by one,
.A2006 BNE $2002
; jump to $2002 if result is not zero,
.A2008 BEQ $2000

This is what the program does: the X register is loaded with $04 which then goes into the frame color (setting the frame color to cyan). X is then decreased by one (X is now $03). As DEX resulted not in $00 but in $03 BNE jumps to $2002, which now sends the value of X (= $03) into the register for the frame color (now, it's red). Now X is decreased once more. DEX has still not reduced X to $00, therefore the program jumps back to $2002 where the value of X (= $02) again is loaded into the frame color. This process continues until DEX has decreased the value of the X register to $00, which is when the processor will automatically set the ZERO-FLAG. But as BNE only jumps on the condition that the ZERO-FLAG is CLEAR (that is, that the last operation resulted in something other than a zero), the program continues to $2008. The following command, BEQ $2000, only jumps if the ZERO-FLAG is set - which must the case, otherwise BNE would have jumped before. So BEQ carries us back to $2000 where the program starts anew. Now what effect does this program have on the screen? Same as this one:

.A2000 LDA #$04
.A2002 STA $D020
.A2005 LDA #$03
.A2007 STA $D020
.A200A LDA #$02
.A200C STA $D020
.A200F LDA #$01
.A2011 STA $D020
.A2014 JMP $2000

The difference is that the program that works with jumps controlled by certain conditions is much shorter. But now let's look at our last program with which we're going to read a table for the first time. First, type this nice little table:

.M3000 06 06 04 04 0E 0E 03 03
.M3008 0D 0D 01 0D 0D 03 03 0E
.M3010 0E 04 04 06 06 00 00 00
.M3018 01

What you did just now was to set up a table of color values which occupies storage locations $3000 to $3018, and which we're now going to use to have the screen change to these colors. The program looks like this:

.A2000 LDX #$19
; number of colors plus one into the X register,
.A2002 LDA $2FFF, X
; loads value of $2fff into the accumulator (table starts at $3000)
.A2005 STA $D020
; value of the accumulator into frame color,
.A2008 STA $D021
; value of accumulator into background colors,
.A200B DEX

; decreases X register by one,
.A200C BNE $2002
.A200E BEQ $2000

Well, how does this program work? It starts with the last value of the table and continues to the first one, loading the values from the table into the registers for frame and background colors. The X register is used to proceed from one value to the next in the table. First, X is loaded with #$19 so that LDA $2FFF,X loads the last value from the table, which will come from storage location $3018 ($2FFF + X, X=$19), into the accumulator. The value of the accumulator is then loaded into the screen colors ($D020 and $D021). Then X is decreased by one, which results in $18 in the second run. Since $18 does not equal $00, the BNE jumps to $2002. At that point, the value from $3017 ($2FFF + $18) is loaded into the screen colors ... and so forth.

The interesting part of this comes when X reaches a value of $01. At this point, the first value of the table is read, which is $3000. This value is now loaded into the screen colors and X is decreased by one - to $00. Logically, the BNE does not jump anymore. If the command had been LDA $3000,X we now wouldn't be able to read the first value of the table. In any case, BNE is ignored because the ZERO-FLAG is set. But the following BEQ jumps to the beginning of the program at $2000 due to the ZERO-FLAG - and we're back in our old familiar closed loop.

In the next part we will dedicate ourselves to finding out if we can achieve some even nicer effects concerning tables and colors - and by the way we'll have the chance to try some great new commands. Hope you're looking forward to it, see you next time!
Assembler course part 3
by Wanja Gayk

Hi guys! The programs we've written so far aren't really all too useful, wouldn't you admit? But now we're going to try some cooler stuff. Today I want to show you how to produce a nice color effect on the screen ... Interested? Okay, start your monitors and let's get going!

For starters we're going to let the first line of the screen change color. To be able to do that it's necessary to know how the C64 handles colors. This is how it works in the text mode: if you look at the screen memory you'll notice that the screen code for the character from the upper-left corner can be found in $0400 ... the next character to the right of it is in $0401 and the next chAracter to the right of that can be found in $0402, and so on. The first sign of the second line is in $0428 ($28 hex = 40 decimal), the one right of it in $0429 and so forth. The system's easy enough to understand, isn't it? You can find the screen codes for each of the characters in the C64 manual. I'd like to illustrate the whole thing on the screen with a simple Assembler program for which we need a new command:

TXA / TAX and TYA / TAY

TXA (Transfer X register to Accumulator), this command copies the value from the X register into the accumulator. To copy a value from the accumulator into the X register you have the command TAX (Transfer Accumulator to X register). These two commands are of course also available for the Y register, where they are called TYA (Transfer Y register to Accumulator) and TAY (Transfer Accumulator to Y register) respectively. The following program will show you their effects:

.A 1000 LDX #$00
.A 1002 TXA 
.A 1003 STA $0400, X
.A 1006 INX 
.A 1007 BNE $1002
.A 1009 RTS

This program loads the values of $00 to $ff into the screen memory. It does a great job in demonstrating which value from the screen memory (screen code) is responsible for which sign. On some older C64's, you won't see anything with this code. The reason is that the older Kernal's internal CLR/Home routine does not fill color memory with the current cursor color, but instead, with the current BACKGROUND color. Any characters output without a specific color (e.g. by POKE or STA) are as invisible as an albino rabbit on a glacier (white on white). Just a few preparations will help solve the problem. Can you imagine which?

This is how the program works: We start the X register at $00 and copy it into the accumulator. Now we load the content of the accumulator into the screen memory at $0400 + X. In the first run X is $00, so that the value of the accumulator (presently $00) is loaded into memory cell $0400. Now the X register is increased by one via the INX. As X does not equal zero, the BNE (Branch if Not Equal) instruction jumps back to $1002 where the value of the accumulator (now $01 thanks to the TXA instruction) is loaded into the screen memory. This process of copying X into the Accumulator and then storing to memory at $0400 + X continues until X = $ff. Again $ff is loaded into the screen memory at $04ff. The next command is INX, which tries to increase X once more. Since it has already reached $ff, the maximum value, X starts again at zero. Now that X has rolled over to Zero, the Zero-Flag is now set, and BNE does not jump anymore. The program proceeds to RTS and stops. RTS is a new command I'd like to explain briefly:

RTS (ReTurn from Subroutine) - Normally subroutines are called up with SYS from BASIC or with JSR from Assembler. We're going to use JSR later on in this part of the course, where the command is going to be explained. Every time you call up a subroutine, the address to which it must return is stored on the processor stack. RTS takes these return addresses from the stack again and jumps back to them. RTS works comparable to RETURN in BASIC. In BASIC the command GOSUB 100 calls up a subroutine. RETURN then returns to the spot where the GOSUB command left off. So if you jump with SYS from BASIC to an Assembler program RTS will return from Assembler to BASIC.

Well, now you've seen the elements the screen is composed of and know which screen codes represent which characters. But where are the colors I promised you so pretentiously? No problem whatsoever, since now we'll come to the color memory - also called color RAM. It starts at $d800 and is composed of the same elements as the screen memory. Each memory cell of the color RAM represents the color of one character found in the corresponding cell of the screen memory. The colors are treated similar to the border ($d020) and background ($d021) colors. Values from $00 up to $0f set the color of any one character. All values higher than these do not make sense as they only produce the same colors, i.e. $0a = light red, $1a = light red, $2a = light red, and so on. As you can see, only the second, low-order digit of the number - the so-called low-nybble (4-bits) - is important. You'll be best able to see that with the help of a little program with which we'll color the first line of the screen. First, let's set up a little color table:

.M 1100 06 04 0e 0a 03 0f 0d 07

.M 1108 01 01 07 0d 0f 03 0a 0e

.M 1110 04 06 0b 0c 0f 01 0f 0c

.M 1118 0b 09 02 08 0a 0f 07 01

.M 1120 01 0d 03 05 0c 0b 0c 0f

And now the program: 

.A 1000 LDX #$00
.A 1002 STX §D020
.A 1005 STX $D021
.A 1008 LDA $1100, X
.A 100B STA $D800, X 
.A 100E INX 
.A 100F CPX #$28
.A 1011 BNE $1008
.A 1013 RTS

What to do: just write anything you like in the first line of the screen in the BASIC editor, press SHIFT+RETURN (so the line will be ignored and you'll get no syntax error warning), start the program with SYS 4096 or from the monitor with G 1000, and you'll have a nicely colored first line.

Now, how does this program work? Well by now you probably know, but here goes: First, we initialize X to $00, and store this zero value to the border and background color registers (convenient, isn't it?). Next, we start a loop, in which we first load a value from our table. The location pointer to will be $1100 + X, which will be $1100 since X is zero on the first run. The value is then stored to Color RAM at location $D800 + X, which will be $D800 on the first run, of course. X is then increased by 1 (INX) and compared against the value $28 (40 in decimal). If X is not equal to $28, the BNE instruction will cause a jump back to $1008. If they are not equal, the instruction fails and falls through to the RTS, where the program terminates.

Each time the BNE branches back to $1008, X is increased by 1 and a value is loaded from the table and stored to Color RAM. On the second pass through the loop, X will be 1, and so the byte copied will be from $1101 to $D801, followed by $1102 and $D802, and so on until X reaches $28.

Okay, so now we have a colored first line, but it would be nice to have some motion, wouldn't it? So how are we going to do it? Let's start like this: let's take the first color value from our table and keep it in a temporary storage location. Then we move the whole table one position to the left. The second color value now takes the place of the first, the third that of the second and so on. The last value changes to the second last and leaves an empty space where we can deposit the value we have stored. Now let's bring the colors back onto the screen, which means copying the color table into the color RAM.

In addition to the above routine we now need another one to be able to cycle the color table. And as it would be much too easy to store the first value if the table in the X or Y register, we'll use two new commands:

PHA / PLA (PusH Accumulator, PulL Accumulator)

PHA pushes the accumulator's content onto the processor stack in the same way that the processor pushes return addresses when executing a JSR or SYS command. PLA pulls the first value from the stack and transfers it to the accumulator.

The stack is used to store jump addresses and other values. It works by the LIFO principle (Last In - First Out). That's to say if you push one byte directly after another onto the stack by use of the PHA and JSR instructions, the first PLA takes the byte that was most recently pushed, from the stack, and places it in the Accumulator. The second PLA pulls off the next most receltny puched byte. Since return addresses are also stored on the stack the number of bytes PULLed from within a subroutine or program must always equal that of those PUSHed within that same subroutine or program, otherwise you'll get a magnificent crash!

But now let's look at a program that demonstrates the effects of the PHA and PLA commands. This program starts off at the address $1014 - directly following the other program.

.A 1014 LDA $1100
.A 1016 PHA
.A 1017 LDX #$00
.A 1019 LDA $1101, X
.A 101C STA $1100, X
.A 101F INX
.A 1020 CPX #$27
.A 1022 BNE $1019
.A 1024 PLA 
.A 1025 STA $1127
.A 1028 RTS


What exactly is the effect of this little routine? It takes the first value from the table (at $1100) and pushes it onto the stack. Then it returns the X register value to $00, takes the second value (from $1101 + X) and puts it in the place of the first (to $1100 + X). X is increased by one and a check is made to see if X has already reached the number of table values (minus one, as the last one has been exchanged for the first one, so you don't need it anymore). If this is not the case, the loop is repeated. Otherwise, BNE will not jump and the program will continue at $1024 where the byte last pushed onto the stack will be pulled off and put in the place of the last value of the table. Now we're finished - you've cycled through the table once. You only have to display it on the screen with the routine you wrote at $1000 (first program). Of course you could write a similar BASIC program looking like this:

10 SYS 4096: SYS 4116: GOTO 10

But this is an Assembler course, isn't it? So for testing purposes let's write a little routine at $1200 just to use it for demonstration. We're going to need one new command for it.

JSR (Jump to SubRoutine)

JSR jumps to a subroutine elsewhere in the computer's memory. JSR $1000, for example, would jump to the program that starts at memory cell $1000. The command has the same effect as GOSUB in BASIC - the equivalent of BASIC's RETURN command (to end a subroutine) is a simple RTS instruction in Assembler, as you already know. That's the reason why RTS jumps back to BASIC every time you call up the machine routine from BASIC with SYS. The command SYS is nothing but a JSR command disguised by the BASIC interpreter.

But now let's finally come to the program. It doesn't return to BASIC because it doesn't have an RTS command at the end (it's an infinite loop, anyway):

.A 1200 JSR $1000
.A 1203 JSR $1014
.A 1206 JMP $1200


A really simple program: it jumps to the routine that displays the color table (JSR $1000) and then to the program that cycles the table ($1014). After that it jumps back to its own start where it branches again to the display routine, forming an infinite loop.

You can easily start the effect on the screen with G1200 or in BASIC with SYS 4608 - to abort you've just got to press RUN / STOP-RESTORE - experts like you already know that of course. You'll see that things run nearly a little too fast (well, yes - there's nearly no stopping our dear little C64 with Assembler programming). Therefore I'm going to show you in the next part how to synchronize the whole thing with the scanning beam that is responsible for the display. Then you'll get really close to the cool stuff: the INTERRUPT. Okay, stay hot for next time!
Assembler course part 4
by Wanja Gayk

Hi! I hope you've brought some time with you because now we are slowly arriving at the point where programming starts being fun. The last part was about bringing color-effects onto the screen by copying a color-table into the color-ram, moving this chart to the next position, inserting the color, that comes out at one end, at the other end and finally copying our new color-table into the color-ram again.

So we had a closed loop. Now we want to see the whole thing in a reasonable, regular speed and... well, we want to show off a bit and to bring this effect into the background of a Basic-program. Hold it, hoolllld it!! You can't have THREE Wishes! That's impossible!

[Kendra, Wanja, Guenther: The only thing I could think of to replace this slogan was a variation on the traditional Genie-in-the-bottle. Old stories say that if you find a genie in a lamp or bottle and release him, he'll grant you a wish. I hope this works for you. -Nate]

Actually, it isn't all that impossible, and we certainly don't need any genie to help us out. All we need is a normal breadbox C64 and a good machine language monitor!

Also, we need to know what an interrupt is. Then we can confidently forget our closed loop. Don't worry, you haven't learned everything in vain. But what the hell is an interrupt?!?

Now let's imagine the following: The C64 builds up its picture 50 or 60 times every second, which you may be able to see depending on your TV or monitor. In additon, the computer carries out several service-routines that, for example, make the cursor flash or wait for an input from the keyboard. To do this, a chip interrupts the C64's actual work from time to time - so this is simply what "interrupt" stands for. Now the trick is to latch your own programs to these C64 service-routines, or in some cases, to replace them completely. At the beginning of each interrupt the computer gets the start address of the service-routine out of the addresses $0314 and $0315 . There we normally have $31 and EA, respectively. Inside the storage addresses are always stored exactly backward from the you would read them. The order is lowbyte, highbyte. That means that during every interrupt the computer jumps to the address $EA31 ($EA is the highbyte and $31 the lowbyte). You might be confused a bit now, but don't worry, things will become clear soon. The addresses $0314 and $ 315, from where the start-adresses for the service-routines come, are called an interrupt vector.  The word vector actually means "an entity describing both direction and distance from a starting point." In the case of a computer, it means the direction and distance from the vector location itself, to jump to. Now we latch our programs to these routines by directing the interrupt vector to the start-address of our own program.

If we wanted to put our interrupt routined at address $1100, we simply redirect the vector by writing the $11 $00 into the interrupt vector at $0314/ $0315, as shown here:

.A1000 LDA #$00
;low byte
.A1002 STA $0314
.A1005 LDA #$11
;high byte
.A1007 STA $0315
.A100A RTS

But wait! What if the computer wants to carry out an interrupt while we're in the middle of changing the two-byte vector? "Vvrroooomm... SCREEEECH!!! Crash!!" - The computer comes to a screeching halt. Of course we want to avoid that, so we must forbid the computer from carrying out any interrupts while we are working with the interrupt vector. For this we need 2 new commands:

SEI - SEt Interrupt disable flag: This command turns interrupt off.
CLI - CLear Interrupt disable flag: This command restores interrupts.


So our new program is:

.A1000 SEI
.A1001 LDA #$00
.A1003 STA $0314
.A1006 LDA #$11
.A1008 STA $0315
.A100A CLI
.A100B RTS

If you start this now (with SYS 4096 in BASIC or g 1000 in your monitor) there will still be a crash, because there isn't yet a program at $1100 (where we have directed the interrupt to)! Patience. In fact, we already have half of it. Now we will latch a small program to the interrupt, and to definitely avoid a crash we will jump from the end of our program immediately into the C64's own service routine again. Now back to work! Let's write a small program to $1100:

.A1100 LDA #$07
.A1102 STA $0400
.A1105 LDA #$0F
.A1107 STA $0401
.A110A LDA #$36
.A110C STA $0402
.A110F LDA #$34
.A1111 STA $0403
.A1114 LDA #$21
.A1116 STA $0404
.A1119 JMP $EA31

Now, enter lines $1000-$1010 from above, and this second part (lines $1100-$1119), leave the monitor with X and divert the interrupt with SYS 4096. Now, have a look into the upper left corner of the screen - try to delete or overwrite what you see there (with Shift-Clear/Home or the delete key, perhaps).

IF your machine doesn't show anything at the top of the screen, try pressing Home, followed by several spaces (or anything else). This oddity is due to a bug in some very old C64 ROMs, which would cause characters to be invisible is thier color locations aren't properly set up. The Home + Spacebar thing takes care of this.

As you can see, the small "GO64!" in the corner just cannot be killed... unless you boldly press Run/Stop-Restore. Why? Because every time the computer produces an IRQ (interrupt), it jumps into our routine which writes the string "GO64!" into the corner and jumps into the system- routine that would have originally been carried out. So the C64 acts as if nothing had happened. Bingo! Now it should be clear, why we mustn't program a closed loop into an interrupt, otherwise the computer would be trapped inside the
interrupt, constantly trying to start a new interrupt during this endless loop. After a few seconds, it would be confused and crash.

Alright: An interrupt routine must never contain a closed loop and must be terminated properly -- that means by jumping into the system routines!! If everything is clear so far, we are going to the next step. If not, go back and re-read the text a couple of times - it can sometims take a while for the idea to really "sink in".

There are a few different kinds of interrupts. The most important interrupt for the programer is the screen, or "raster" interrupt. What you have to know is, that the screen is built up one raster line at a time, from top to bottom. With a screen interrupt we can decide on which screen line the interrupt shall begin. On NTSC machines, screen line No. 1 is below the bottom line of the text display, immediately after the "last" line (line 262). On PAL machines, line No. 1 is inside the upper screen frame, while last line (312) is down at the bottom and inside the screen as well, but out of our sight. As soon as the raster beam finishes the last line, it immediately begins again at the first line.

We have to tell the computer which kind of interrupt we want and where. For this the VIC has certain registers. I guess it's the best I show you with a small program again:

.A1000 SEI

; disable interrupts
.A1001 LDA #$00
; "bend" interrupt vector to $1100 
.A1003 STA $0314
.A1006 LDA #$11
.A1008 STA $0315
.A100B LDA #$60
.A100D STA $D012
; interrupt starts in screen line $60
.A1010 LDA #$1B
.A1012 STA $D011
; text screen on and bit 7 is clear.
.AD011 LDA #$F1
.A1017 STA $D01A
; start raster interrupt
.A101A CLI        
; re-enable interrupts
.A101B RTS

Line $1010 is particularly interesting: Why must we clear bit 7 in $D011? The reason is that the screen has more than 256 lines. Since one byte (namely $D012) can only represent a number from $00 to $FF (255, decimal), one additional bit is provided in $D011. If the screen beam is beyond or below the 256 mark, this additional bit will take care to represent this.  You may know this from playing around with sprites, which are also difficult to move to the right margin of the screen. As long as bit 7 in $D011 is cleared, the C64 knows that you want to start the interrupt in the area of $00-$ff and not further at the bottom. As long as you don't have any crazy ideas, just stay at the given value.

But what do we gain from this interrupt? Well, as we know the interrupt always starts in the line that we want. This gives us a quite regular timing (at least regular enough for our means), compared to our usual timer-interrupt. And with this we can achieve our original aim: a color-effect without jerkiness or jitters. We use the mentioned routine to initialise our interrupt. What is still missing is our own Interrupt-routine for the color-effect. Let's remind ourselves: We take a routine to copy the color- table into the color- ram and another to revolve/ rotate the color-table. And, in an interrupt, we can not have ANY closed or infinite loops!

I still have to make one peculiarity clear to you: We have to access register $D019 once while writing, so that a screen interrupt is recognized as finished. We use the standard way to do this, as the SuperCPU is a bit picky here, and simply write anything into $D019, e.g. a $01.

Now our program looks like this:

Interrupt- routine:

.A1100 JSR $1180
; call the "copy table" subroutine, below.
.A1103 JSR $1200
; call the "revolve table" subroutine
.A1106 LDA #$01
.A1108 STA $D019
; clear interrupt- register (interrupt finished)
.A110B JMP $EA31
; continue with regular system- routine

Subroutine: "copy table into colour-ram":
.A1180 LDX #$00
; X-register to $00
.A1182 LDA $1300,X
; value from table + fetch X
.A1185 STA $D800,X
; value into colour-ram + write X
.A1188 INX

; increase X by 1
.A1189 CPX #$28
; compare X with $28 (decimal 40)
.A118A BNE $1182
; if it's not a match, jump to $1182
.A118C RTS

; (end of subroutine)
Subroutine: "revolve table":
.A1200 LDA $1300
; load value to start into table
.A1203 PHA

; move onto stack
.A1203 LDX #$00
; X-register to $00
.A1206 LDA $1301,X
; value from table + read X 
.A1209 STA

; value into table + write X 
.A120B INX

; increase X by1
.A120C CPX #$27
; compare X with $27 (dec.39)
.A120E BNE $1206
; if unequal, jump to $ 1206
.A1210 PLA

; take value from stack
.A1211 STA $1327
; and write it into end of table
.A1214 RTS

; (end of subroutine)

Color table:

.M1300 06 04 0e 0a 03 0f 0d 07
.M1308 01 01 07 0d 0f 03 0a 0e
.M1310 04 06 0b 0c 0f 01 0f 0c
.M1318 0b 09 02 08 0a 0f 07 01
.M1320 01 0d 03 05 0c 0b 0c 0f

You can start the program if you leave the monitor and enter a normal SYS 4096. Now you will see, that you have got a colour- effect in the top line of the screen that works at the same time as the regular BASIC interpreter, and looks good as well! There we are! In our next installment, I will show you how you can slow the effect down without a delay loop, because long delay loops are taboo for the interrupt!! Last but not least I want to show you some other simple, but nice effects using the VIC- registers $D011 and $D016. There's still so awfully much to be discovered... so see you!

Thus, we have to think of another way to slow down effects that are just too fast. The simplest method is by running the program (or part of it) only every other interrupt. By doing this will cut the speed exactly in half.  As before, we will need an interrupt initialization routine that we can use for all following programs. Here is our code:

$1000 SEI

; disable interrupt

$1001 LDA #$00
;r edirect interrupt vector to $1100

$1003 STA $0314

$1006 LDA #$11

$1008 STA $0315

$100B LDA #$FB

$100D STA $D012
; interrupt starts at raster line $FB

$1010 LDA #$1B

$1012 STA $D011
; text screen "on" and at the same time clearing bit 7 in $D011

$1015 LDA #$F1

$1017 STA $D01A
; activate raster line interrupt

$101A CLI

; enable interrupt

$101B RTS
Now we will create a simple effect Let's say we want to make the screen shake a little. To do

so, we use the VIC horizontal scroll register at $dO16.This register allows one to scroll the

entire display (or a portion of it) up to eight pixels horizontally without using any

significant processor power.
· IRQ Main Routine —
$1100 JSR $1110

$1103 LDA #$01
; finished

$1105 STA $D019

$1108 JMP $EA31
; scroll routine confirm interrupt as and on with the original ROM routine

In our last Installment we discussed writing an interrupt service routine. The only problem was that the effect ran too fast and that we are not allowed to integrate any delay loops in the interrupt, because we could "overload" an interrupt. Overloading in this context, means that the interrupt routine takes so long, that the computer will try to start a new interrupt while the old one is still being serviced. This shows up as a rather displeasing flicker, distorted music or some other stuff... and we don't want that to happen, of course.

- Hardware Scroll -

$1110 DEC $1200
; decrease delay counter
$1113 BEQ $1115
; if result = 0, branch to $1115

$1114 RTS

; otherwise immediate return from subroutine

$1115 LDA #$02
; $02 in Accumulator (start value for counter)

$1117 STA $1200
; and write to delay counter

$111A LDY $1201
; load table counter (index) to y- register

$111D LDA $1202,Y
; and retrieve scroll value from table + Y to accumulator

$1120 ORA #$08
; set bit 3 (for 40- columns mode) in accumulator

$1122 STA $D016
; and write acuumulator to scroll register

$1125 INY

; increase y- register by 1

$1126 CPY #$13
; and compare it to $13

$1128 BNE $112C
; branch to $102c if not equal

$112A LDY #$00
; all through, reset yregister to zero

$112C STY $1201
; and write to table counter

$112D RTS

; Return from sub table counter, delay counter and

;& scroll value table ($13 values) –

m $1200 02 00 01 02 03 04 05 06

m $1208 07 07 07 07 06 05 04 03

m $1210 02 01 00 00 00

Now, are you confused? I hope that this will be over soon. This whole assembly works like this: in interrupt init I choose the raster line in which the interrupt shall be triggered, in this case, raster line $FB, which is in the lower screen border. Every interrupt I call the scroll routine. At that time the electron beam is defintely still in the lower border and we remember when it leaves the lower border, it automatically returns to the top - but our routine won't take that long. This way we make

absolutely sure that the scroll value is already set before the screen is drawn, so that we don't

have any "twisting" on the screen, where a new scroll value replaces the old. But allright, these

are details that you will surely investigate more closely later. Back to the scroll routine.  It starts immediately with our "delay". The principle is quite simple: I take a counter that I keep decreasing by 1 and as long as it isn't zero, I leave the subroutine without doing anything. But once the counter reaches zero, I reset it to its initial value (the $02 in $1200 in the table) and execute my real routine. This starts by getting an index on the table and then the scroll value from table + index. 
The next instruction is new:
ORA #$xx - Perform a Logical OR operation on the Accumulator against the value #$xx,

leaving the result in the Accumulator. With ORA, one can set single bits as desired (See

illustration 1 - Logical Operations). Other logic instructions include AND #$xx (Logical AND

operation) and EOR #$xx (Logical Exclusive- OR operation).  Using the ORA instruct ion, I set the bit which activates 40 column mode. Keep in mind that register $dO16 has several functions (refer to illustration 2 - Registers $d011 and $dO16).  Anyway, now we have the final value that we write to register $dO16. The following program lines increase the table counter by 1 and, if it exceeds the number of table entries, reset it to zero. Then the subroutine ends, and we find ourselves in the main IRQ routine, which jumps to the normal interrupt routine found in the Kemal ROM.  You should take another long look at it all, so that you place everything correctly. If you want to experiment a bit, try to do the same thing with register $d011, so as to make the screen go up and down. Or maybe even both movements together? I don't want to limit your adventurous spirit - go at it! In the next part, there will once again be something basic about adressing methods.  Maybe that will seem somewhat boring at first sight, but everything will advance so much easier afterwards.

Last time I promised you we'd take a look at the different addressing modes. I'm afraid this is going to be a little bit dull, but you should read it by all means, as it'll be important later when we

concentrate on the essentials, namely registers and effect routines.

by Wunja Ggvk index, i.e. LDA $1000,~. But now it'll become changing the vector used by the LIST

more tricky, as we want to access a memory command to point to the routine that prints a

cell indirectly - the magic word being SYNTAX ERROR.

L et's begin with the different ways of "addressing" the memory, how we read,

write, or change the contents of memory cells.

The most important addressing mode is called

DIRECT (or ABSOLUTE) ADDRESSING.

We already know this mode; 'direct' means we

access a memory cell DIRECTLY, without any

indirections, e.g.,

LDA address

STA address

We read the value at an address in the 64's

address space into the accumulator or write the

value in the accumulator into the byte at an

address. Of course, the same holds for the X

and Y registers, meaning "LDX address",

"STX address", "LDY address" and "STY

address" are legal, direct addressing

commands. Another kind of direct addressing

is called

Direct Zeropage Addressing

which is exactly the same as direct addressing,

with the difference of the address being in the

interval from $00 to $ff, and so the commands

only need a single parameter byte to hold the

address. For instance, we may write LDA $fa

instead of LDA $OOfa.

Direct Indexed Adressing

1 This is another addressing mode that we've

already seen:

loads the byte at address+x into the accu. For

1 instance, if address equals $1000, and x holds

3, the command "LDA $1000,x" would load

1 the value at address $1003 into the accu.

Equally to the X register, you can use Y as an

Indirect Addressing

The most simple kind of indirect addressing we've already seen is with the JMP command.

While a direct "JMP address" jumps to the address directly (as the name tells), we may

perform an indirect jump by writing: JMP (address)

The address in the parentheses is the location in memory where the JMP command finds its

actual target address. For instance, if you want to jump to different addresses, you don't have

to change your code but just the target address stored in memory at the address in brackets.

Because this kind of parameter does not directly specify an address but rather points to

one, we usually call this a VECTOR, and we normally write "JMP (vector)" instead of "JMP

(address)". This example illustrates how vectors work:
Ida #$34

sta $3000

Ida #$I2

sta $3001
In this example, "JMP ($3000)" jumps to the address $1234, which the vector at $3000

represents in the "LOWBYTE, HIGHBYTE" format. What's it good for, you may ask? For

instance, you may now work with vector tables that you can change just as needed. The

operating system uses this method to call system routines like the LOAD command,

which is always called through the vector at $0330. A fastload utility changes this vector to

its own load routine. Then if you issue a LOAD command with the fast loader installed, the

operating system will jump through the vector at $0330 and is taken to the new load routine

instead of the original one. Another application is a simple LIST protection you can get by

We can even extend the idea of indirect addressing by combining it with indexing.

Again we can use the X and Y registers for indexing, but this time they perform different

functions:
Y-Indexed lndirect Zeropage Addressing sounds incredibly complicated, but it's not.  ZEROPAGE simply means the vector is located at an address below $0 100 in the range between $00 and $ff, and so it can be represented as an operand to a command by just a single byte (the highbyte of zeropage addresses is always zero).  (Who wrote this stuff'?!) The notation for this type of addressing is as follows:  i.e., we load the byte at the address pointed at by the vector at $fa (and $fb) plus the value of the Y register. The following example should clarify how it works:
Ida #$34

sta $fa

Ida #$I2

sta $fb

ldy #$03

Ida ($fa) ,Y
The last command loads the byte at address $1234 (given by the vector at $fa) plus $03

(value of Y register). and so, as a result, the rn

byte at address $1237 is read into the accumulator. But what's the use of such a

complicated command? Well, it's more than 4

just useful; for example, an "LDA $1000,y"

allows us to access 256 bytes ($1000 through $]Off) with one single address. We can do the same with an "LDA ($fa),yV, but additionally, the vector at $fa may point ANYWHERE in the whole address space between $0000 and $ffff.  We can access the whole memory with this

